BASIC Programs for the Evaluation of Wavenumbers and for Solving Polynomials

K. Janardhanam and S. V. J. Lakshman*

Sri Venkateswara University, Tirupati 517502, India

Received February 26, 1992

BASIC programs useful for the conversion of wavelengths into vacuum wavenumbers and for solving polynomials are presented. The latest (1972) dispersion formula of Peck and Reeder for the refractive index of air is used in the evaluation of the wavenumbers. In both programs, the Birge-Vieta method is used for solving the polynomials.

INTRODUCTION

In all spectroscopic work, the measured wavelengths of the observed spectral lines or bands must be converted into vacuum wavenumbers for the evaluation of spectroscopic parameters. For the conversion of wavelengths into vacuum wavenumbers, Kayser's[1] tables were used in earlier days and NBS wavenumber tables[2] more recently. Wavenumbers are evaluated using the relation $\nu = 1/n\lambda$, where ν is the wavenumber, λ is the wavelength, and n is the refractive index of air at that particular wavelength.

In the preparation of the NBS wavenumber tables, Coleman et al. used the following formula by Edlen[3] for the correction of the refractive index of air:

$$n = 1 + 6432.8 \times 10^{-8} + \frac{2\,949\,810}{146 \times 10^{8} - \nu^2} + \frac{25\,540}{41 \times 10^{8} - \nu^2} \tag{1}$$

where ν is the wavenumber (in cm^{-1}).

In the early days when computer time was very expensive, it was necessary to consult the large volumes of NBS wavenumber tables for the conversion of wavelengths into wavenumbers. Now, any scientist can afford to buy a reasonably-priced pocket computer and use it for computational work such as the evaluation of wavenumbers.

Table I. BASIC Program for Evaluation of Wavenumbers

```
100 REM CALN OF WAVENUMBER
110 DIM A(5), M(5), R(5)
120 INPUT "W.LENGTH=";X
130 Y=X/1000
140 A(1)=-1.000272608
150 A(2)=-3.298622625/Y^2
160 A(3)=3.299368771/Y^2
170 A(4)=1.922234267/Y^4
180 A(5)=-1.922389208/Y^4
190 G=1
200 M=1
210 M(1)=G+A(1)
220 R(1)=1
230 FOR K=2 TO 5
240 M(K)=A(K)+M(K-1)*G
250 R(K)=M(K-1)+R(K-1)*G
260 NEXT K
270 IF ABS (R(5))>0.0001 THEN 300
280 G=G+0.1
290 GOTO 210
300 H=G-M(5)/R(5)
310 IF ABS (H-G)<=0.0001 THEN 380
320 IF M>20 THEN 360
330 M=M+1
340 G=H
350 GOTO 210
360 PRINT "FAILS TO CONVERGE"
370 GOTO 400
380 Z=100000/(H*Y)
390 PRINT "WAVENUM=";Z
400 GOTO 120
```

Note: 1. The DIM statement in line number 110 is to be deleted for pocket computers like CASIO PB 110.

2. Wavelength X is to be fed in Angstroms.

0095-2338/92/1632-0369$03.00/0

Table II. Comparison of Wavenumbers (in cm^{-1}) Calculated Using Equations 3[a] and 1[b]

wave-length, Å	wave-number using eq 3	wave-number using eq 1	wave-length, Å	wave-number using eq 3	wave-number using eq 1
2100	47 603.9372	47 603.9373	6600	15 147.3316	15 147.3315
2200	45 440.3586	45 440.3590	6700	14 921.2536	14 921.2536
2300	43 464.8753	43 464.8758	6800	14 701.8249	14 701.8249
2400	41 653.9860	41 653.9865	6900	14 488.7564	14 488.7564
2500	39 987.9460	39 987.9464	7000	14 281.7755	14 281.7755
2600	38 450.0464	38 450.0467	7100	14 080.6250	14 080.6251
2700	37 026.0525	37 026.0527	7200	13 885.0620	13 885.0620
2800	35 703.7624	35 703.7625	7300	13 694.8568	13 694.8568
2900	34 472.6566	34 472.6566	7400	13 509.7923	13 509.7923
3000	33 323.6180	33 323.6180	7500	13 329.6627	13 329.6627
3100	32 248.7057	32 248.7056	7600	13 154.2734	13 154.2734
3200	31 240.9709	31 240.9708	7700	12 983.4396	12 983.4396
3300	30 294.3073	30 294.3071	7800	12 816.9861	12 816.9861
3400	29 403.3267	29 403.3265	7900	12 654.7466	12 654.7466
3500	28 563.2567	28 563.2565	8000	12 496.5630	12 496.5631
3600	27 769.8550	27 769.8548	8100	12 342.2852	12 342.2853
3700	27 019.3380	27 019.3378	8200	12 191.7703	12 191.7703
3800	26 308.3202	26 308.3201	8300	12 044.8821	12 044.8822
3900	25 633.7635	25 633.7634	8400	11 901.4913	11 901.4914
4000	24 992.9334	24 992.9333	8500	11 761.4744	11 761.4744
4100	24 383.3623	24 383.3621	8600	11 624.7136	11 624.7137
4200	23 802.8174	23 802.8173	8700	11 491.0968	11 491.0968
4300	23 249.2738	23 249.2736	8800	11 360.5167	11 360.5167
4400	22 720.8905	22 720.8904	8900	11 232.8709	11 232.8710
4500	22 215.9903	22 215.9902	9000	11 108.0618	11 108.0618
4600	21 733.0417	21 733.0416	9100	10 985.9956	10 985.9956
4700	21 270.6436	21 270.6435	9200	10 866.5830	10 866.5831
4800	20 827.5117	20 827.5116	9300	10 749.7385	10 749.7385
4900	20 402.4663	20 402.4662	9400	10 635.3800	10 635.3800
5000	19 994.4224	19 994.4224	9500	10 523.4290	10 523.4290
5100	19 602.3799	19 602.3799	9600	10 413.8103	10 413.8104
5200	19 225.4157	19 225.4156	9700	10 306.4518	10 306.4519
5300	18 862.6763	18 862.6762	9800	10 201.2843	10 201.2843
5400	18 513.3714	18 513.3714	9900	10 098.2413	10 098.2414
5500	18 176.7683	18 176.7683	10000	9997.2592	9997.2593
5600	17 852.1866	17 852.1865	11000	9088.4199	9088.4200
5700	17 538.9934	17 538.9934	12000	8331.0533	8331.0534
5800	17 236.5999	17 236.5999	13000	7690.2043	7690.2043
5900	16 944.4569	16 944.4569	14000	7140.9049	7140.9049
6000	16 662.0518	16 662.0518	15000	6664.8452	6664.8453
6100	16 388.9058	16 388.9058	16000	6248.2929	6248.2930
6200	16 124.5709	16 124.5708	17000	5880.7467	5880.7467
6300	15 868.6274	15 868.6274	18000	5554.0389	5554.0389
6400	15 620.6820	15 620.6820	19000	5261.7213	5261.7213
6500	15 380.3657	15 380.3657	20000	4998.6354	4998.6355

[a] This paper. [b] Ref 2.

Taking into account the dependence of refractive index on water vapor, carbon dioxide, pressure, and temperature of the medium of propagation of light, Edlen[4] suggested that in the wavelength region of 6440–3985 Å, the refractive index (n) and wavenumber (ν in cm^{-1}) are better represented by the formula

$$n = 1 + 8342.13 \times 10^{-8} + \frac{2\,406\,030}{130 \times 10^{8} - \nu^2} + \frac{15\,997}{38.9 \times 10^{8} - \nu^2} \tag{2}$$

Peck and Reeder[5] suggested that the following dispersion formula can better fit experimental results from the far infrared region down to 1850 Å:

$$n = 1 + 8060.51 \times 10^{-8} + \frac{2\,480\,990}{132.274 \times 10^{8} - \nu^2} + \frac{17\,455.7}{39.329\,57 \times 10^{8} - \nu^2} \tag{3}$$

where n is the refractive index and ν is the wavenumber (in cm^{-1}).

COMPUTATIONAL PROCEDURE

It is convenient to write eq 3 as

$$n = A + \frac{B}{C - \nu^2} + \frac{D}{E - \nu^2} \tag{4}$$

where

$$A = 1 + 8060.51 \times 10^{-8}$$

$$B = 2\,480\,990$$

$$C = 132.274 \times 10^{8}$$

$$D = 17\,455.7$$

$$E = 39.329\,57 \times 10^{8}$$

Substituting $\nu = 1/n\lambda$ in eq 4 and writing the equation in the form of a polynomial, we obtain

$$n^5 + A(1)n^4 + A(2)n^3 + A(3)n^2 + A(4)n + A(5) = 0 \tag{5}$$

where

$$A(1) = -(ACE + BE + DC)/CE$$

$$A(2) = -(C + E)/(CE\lambda^2)$$

$$A(3) = (B + D + AE + AC)/(CE\lambda^2)$$

$$A(4) = 1/(CE\lambda^4)$$

$$A(5) = -A/(CE\lambda^4)$$

Substituting the values of A, B, C, D and E in the above expressions, we obtain

$$A(1) = -1.000\,272\,608$$

$$A(2) = -3.298\,622\,625 \times 10^{-10}/\lambda^2$$

$$A(3) = 3.299\,368\,771 \times 10^{-10}/\lambda^2$$

$$A(4) = 1.922\,234\,267 \times 10^{-20}/\lambda^4$$

$$A(5) = -1.922\,389\,208 \times 10^{-20}/\lambda^4$$

Equation 5 is then solved by the Birge-Vieta[6] method to get the value of n for a given wavelength λ.

PROGRAM FOR EVALUATION OF WAVENUMBERS

Srinvivas and Lakshman[7] have provided a BASIC program useful for conversion of any wavelength into its wavenumber in about 30 s. But this program is suitable for computers with large RAM area and which have provision for DIM (DIMENSION) statement. Moreover, they have made use of the old formula of Edlen, i.e., eq 1.

This paper presents a revised program making use of the latest dispersion formula of Peck and Reeder, i.e., eq 3. The number of iterations is set as 20, and the limit of accuracy is fixed at 0.0001.

The BASIC program written is given in Table I. The computation time for the calculation of wavenumber for a

Table III. BASIC Program for Solving Polynomials

```
100 REM CALN OF ROOTS OF A POLYNOMIAL
105 DEFM2
110 INPUT "ORDER OF POLY=";A
120 DIM A(A), M(A), R(A), W(A)
140 FOR I=1 TO A
150 INPUT "COEF=";A(I)
160 NEXT I
170 FOR J=1 TO A-1
180 L=A+1-J
190 G=1
200 M=1
210 M(1)=G+A(1)
220 R(1)=1
230 FOR K=2 TO L
240 M(K)=A(K)+M(K-1)*G
250 R(K)=M(K-1)+R(K-1)*G
260 NEXT K
270 IF ABS (R(L))>0.0001 THEN 300
280 G=G+0.1
290 GOTO 210
300 H=G-M(L)/R(L)
310 IF ABS (H-G)<=0.0001 THEN 380
320 IF M>20 THEN 360
330 M=M+1
340 G=H
350 GOTO 210
360 PRINT "FAILS TO CONVERGE"
370 GOTO 470
380 W(J)=H
390 FOR K=1 TO L-1
400 A(K)=M(K)
410 NEXT K
420 NEXT J
430 W(A)=-A(1)
440 FOR J=1 TO A
450 PRINT "ROOT (";J;")= ";W(J)
460 NEXT J
470 GOTO 110
```

```
Note: 1. The DIM statement in line number 120 is to be deleted
         for pocket computers like CASIO PB 110.

      2. The DEFM statement in line number 105 is to be deleted
         for computers like SHARP EL-5500 III, but is necessary
         for computers like CASIO PB 110 to expand the number of
         variables required for solving 4th and 5th order
         polynomials. For solving higher order polynomials, the
         number of variables should be further expanded and
         appropriate variable names are to be used.
```

given wavelength is found to be about 4 s.

As an example, let us calculate the wavenumber for the wavelength of 2500 Å. When the computer asks for 'W.LENGTH', we key in 2500. After a lapse of 4 s, we get the value of the wavenumber (in cm^{-1}) as 39 987.9460.

The wavenumbers evaluated for a few wavelengths are presented in Table II. As is seen, the present wavenumber values are found to widely vary in the fourth decimal place as compared to the values given by Coleman in the shorter wavelength region than in the longer wavelength region.

PROGRAM FOR SOLVING POLYNOMIALS

As the calculation of the value of n is nothing but solving a fifth-order polynomial, the same program with some modifications can be used for finding the real roots of a polynomial of any order. This program is given in Table III. It can give the roots of second- or third-order polynomials in less than 3 s while it takes about 20 s for solving a polynomial of the fifth order. The same program can be used to solve polynomials of higher order by suitably expanding the number of variables in the pocket computer and using the appropriate variable names.

A polynomial of nth order is written as

$$X^n + A(1)X^{n-1} + A(2)X^{n-2} \ldots + A(n-1)X + A(n) = 0$$

The coefficients $A(1)$–$A(n)$ are to be fed to the computer to obtain the n roots of the polynomial.

As an example, let us consider the fifth-order polynomial:

$$X^5 - 4X^4 - 13X^3 + 52X^2 + 36X - 144 = 0$$

When the computer asks for the order of the polynomial, we key in '5'. When it asks next for the coefficients, we key in the values of the coefficients in the following sequence:

−4
−13
52
36
−144

After a lapse of about 20 s, the five roots of the polynomial are displayed as follows:

```
ROOT (1) =  2.0000
ROOT (2) =  4.0000
ROOT (3) = -3.0000
ROOT (4) =  3.0000
ROOT (5) = -2.0000
```

CONCLUSION

Several computations like those above were made on pocket computers like Sharp EL-5500 III, Casio PB 110, and Casio 720-P, and the results on all of them have been found quite satisfactory.

ACKNOWLEDGMENT

We express our grateful thanks to Dr. G. W. A. Milne for bringing to light the latest work on the refractive index of air.

REFERENCES AND NOTES

(1) Kayser, H. *Tabelle der Schwingungszahlen*; Meggers, W. F., Ed.; Edwards Brothers, Inc.: Ann Arbor, MI, 1944.
(2) Coleman, C. D.; Bozman, W. R.; Meggers, W. F. *Table of Wavenumbers, Vols. I and II*; National Bureau of Standards: Washington, DC, 1960; Monograph 3.
(3) Edlen, B. The Dispersion of Standard Air. *J. Opt. Soc. Am.* **1953**, *43*, 339–344.
(4) Edlen, B. The Refractive Index of Air. *Metrologia* **1966**, *2*, 71–80.
(5) Peck, E. R.; Reeder, K. Dispersion of Air. *J. Opt. Soc. Am.* **1972**, *62*, 958–962.
(6) Kunz, K. S. *Numerical Analysis*; McGraw-Hill Book Co.: New York, 1957.
(7) Srinivas, S.; Lakshman, S. V. J. BASIC Program for Evaluation of Wavenumbers from Wavelengths. *Int. Lab.* **1989**, *14* (May/June), 14–16. Also *Am. Lab.* **1988**, *23* (Sept), 23–25.